DECUS
PROGRAM LIBRARY

DECUS NO.	8-650
TITLE	AMIPED - AUTOMATED MEDICAL INTERVIEW WITH PEDIATRIC DATA FILES
AUTHOR	David C. Mauger
COMPANY	University of Auckland Auckland Hospital Auckland, New Zealand
DATE	August 28, 1973
SOURCE LANGUAGE	OS/8 BASIC

AUTOMATED MEDICAL INTERVIEW WITH PEDIATRIC DATA FILES

by

David C. Mauger, M.B.
Department of Pediatrics,
School of Medicine,
University of Auckland

Introduction

The package consists of a program, DRIVER, a segment of it, PRINT, and two data files, QTEXT and PTEXT. It is designed to administer a series of questions in an interactive, branching manner, to record and print a summary of the answers, and to generate a file of these for later reference.

The questions supplied are of a pediatric medical nature, and are intended to relieve the doctor of the burden of personally eliciting some of the repetitious and standardised parts of the pediatric history, but the programs could administer without change, any series of questions. The questions need not be medical in nature.

Requirements

Any OS/8 system with BASIC. Our configuration comprises 8K core, one DEC tape as the system device, and an ASR33 teletype. Devices are not specified for the files in the programs, the system device being assumed; other users will wish to adapt these to their own configurations, in order to achieve greater efficiency.

The programs store the answers temporarily as an array before creating a tape file. Users with a greater selection of peripherals may wish to modify the programs to create the files as the answers are entered.

Description

Questions are read from file QTEXT and printed on the teletype. The first two characters of the first line of each question are used to define the type of question (primary or modifying) and

and the number of lines of to be read.

A choice of responses (RTEXT) is printed on the teletype immediately after the question. The responses may be any suitable terms, but are commonly No/Don't know/Yes/Don't understand, and a code for this format has been included.

Primary questions can only be answered by No/Don't know/Yes, etc. If a primary question is answered by 'Yes', further related questions (modifying questions) are read from QTEXT and printed. The modifying questions may have No/Yes, etc. or free RTEXT. If the reply to a primary question is "No' or 'Don't know', the modifying questions are still read from the file but not printed. A 'Don't understand' reply causes the teletype bell to ring and the program loop back to re-enter the reply.

The replies are stored as an array. Following completion of the last question text corresponding to all possible replies to a single question is read from PTEXT and one printed, depending on the recorded answer.

The coded numeric replies are then stored as a tape file.

The summary may be generated again from the coded file by running the program PRINT

Operation

1. Load the files to the system (or other) device using PIP

   Recommended file names:- DRIVER. BA
   PRINT. BA
   QTEXT
   PTEXT

2. When the system is loaded and responds with a " . " , type R BCOMP (CR) where (CR) = return key.

3. When the system responds with a " * " , type DRIVER. BA (CR)

4. The program should then compile and start, typing "WHAT IS THE PATIENT NUMBER?" Respond by typing any string up to 6 characters to identify the person to be interviewed, e.g. AB1029 (CR)

5. A series of questions will follow. The first is a dummy, for practise only. The last line of each question is always "YOUR ANSWER IS?" Reply by typing the most appropriate numeric character between 1 and 9 e.g. 3 (CR)

6. When the last question has been answered, a summary will be printed. If the need arises to terminate the program before all of the questions have been administered, type '99' as a reply to any question. A summary will be generated for the questions answered. e.g. in reply to "YOUR ANSWER IS?" type 99 (CR)

The running of program PRINT is as for DRIVER, except that the reply to the question "WHAT IS THE PATIENT NUMBER" must be the name of a file previously created by DRIVER e.g. AB1029 (CR)

Example

Shown below is a sample run showing the initial dialogue, a primary and a modifying question, exit using 99, and the printed summary of the two questions.

```
.R BCOMP
*DRIVER.BA
WHAT IS THE PATIENT NUMBER?JDOE

PRACTISE CHOOSING AN ANSWER.

THE CHILD WE ARE TALKING ABOUT IS:
            1.  A BOY
            2.  A GIRL

YOUR ANSWER IS?

--PRESS THE '1' OR THE '2'  KEY------------
     ----------THEN THE 'RETURN' KEY          OVER HERE
?1
OK, THAT'S FINE, HERE WE GO-----

WHEN THE MOTHER WAS PREGNANT WITH THIS CHILD
WAS SHE TOLD BY DOCTOR OR MIDWIFE
THAT SHE HAD TOXAEMIA OR HIGH BLOOD PRESSURE?

1 NO
2 DON'T KNOW
3 YES
9 DON'T UNDERSTAND

YOUR ANSWER IS?3

WERE ANY DRUGS OR MEDICINES PRESCRIBED TO TREAT
THE TOXAEMIA?

1 NO
2 DON'T KNOW
3 YES
9 DON'T UNDERSTAND

YOUR ANSWER IS?99

THAT WAS THE LAST QUESTION. THANK YOU FOR YOUR HELP
GOODBYE JDOE


PAEDIATRIC SUMMARY          JDOE          08/09/73


*** PRENATAL HISTORY ***

TOXAEMIA:NO DRUGS PRESCRIBED

QUESTIONS ANSWERED 'DONT UNDERSTAND': 0
TOTAL QUESTIONS ANSWERED: 2

.
```

```
DRIVER  BA     1.0      09-AUG-73

90 REM PROGRAM DRIVER
100 REM THIS PROGRAM BRANCHES THRU MULTICHOICE QUESTIONS
110 REM SET VARIABLES TO STARTING VALUES
120 PRINT "WHAT IS THE PATIENT NUMBER";\INPUT N$
130 DIM A(171),T$(60)
140 FOR K=1 TO 170\LET A(K)=0\NEXT K
150 FILE #1:"QTEXT"
160 PRINT\PRINT "PRACTISE CHOOSING AN ANSWER."
170 PRINT\PRINT "THE CHILD WE ARE TALKING ABOUT IS:"
180 PRINT "              1.  A BOY"
190 PRINT "              2.  A GIRL"
200 PRINT
205 PRINT "YOUR ANSWER IS?"\PRINT
210 PRINT "--PRESS THE '1' OR THE '2'  KEY------------"
220 PRINT "     ----------THEN THE 'RETURN' KEY           OVER HERE"
230 INPUT J
240 PRINT "OK, THAT'S FINE, HERE WE GO-----"\PRINT
250 LET K=1\LET J=1\LET Q=1
260 REM READ QTEXT M DEF NO OF LINES
270 REM T DEF TYPE 1=PRIM, 7=MULTI QTEXT, 8=MULTI RTEXT,9=OTHER
280 INPUT #1:T$\IF END #1 THEN 800
290 LET M=VAL(SEG$(T$,1,1))
300 LET T=VAL(SEG$(T$,2,2))
310 LET T$=SEG$(T$,3,60)
320 IF T=1 THEN 360
330 IF C<>1 THEN 370
340 IF T<>8 THEN 360
350 PRINT " 1 ";
360 PRINT T$
370 FOR H=2 TO M
380 INPUT #1:T$
390 IF T=1 THEN 430
400 IF C<>1 THEN 440
410 IF T<>8 THEN 430
420 PRINT H;
430 PRINT T$
440 NEXT H
450 REM READ MULTI RTEXT
460 IF T=7 THEN 280
470 REM YES/NO RTEXT
480 IF T=1 THEN 510
490 IF C<>1 THEN 760
500 IF T=8 THEN 560
510 PRINT
520 PRINT "1 NO"
530 PRINT "2 DON'T KNOW"
540 PRINT "3 YES"
550 PRINT "9 DON'T UNDERSTAND"
560 PRINT
570 PRINT "YOUR ANSWER IS";
580 INPUT A(K)
590 IF A(K)=99 THEN 800
600 IF A(K)<>9 THEN 630
```

```
610 PRINT\PRINT "PLEASE CALL HOSTESS"\PRINT PNT(7)
620 LET D(J)=K\LET J=J+1\GO TO 580
630 REM CHECK FOR ILLEGAL ANSWERS
640 IF T<>8 THEN 660
650 IF A(K)<=M+1 THEN 690
660 IF A(K)<=3 THEN 690
670 PRINT "ANSWER INCORRECT-PLEASE CALL HOSTESS";PNT(7);PNT(7)
680 PRINT;PNT(7)\GOTO 580
690 REM STORE ANSWER TO LAST PRIM QUESTION AS C
700 LET Q=Q+1
710 IF T<>1 THEN 750
720 IF A(K)=3 THEN 740
730 LET C=0\GO TO 750
740 LET C=1
750 PRINT
760 REM K COUNTS THE TOTAL QUESTION NUMBER
770 LET K=K+1
780 REM NEXT QUESTION
790 GO TO 280
800 CLOSE #1
810 PRINT\PRINT "THAT WAS THE LAST QUESTION. THANK YOU FOR YOUR HELP"
820 PRINT "GOODBYE ";N$
830 PRINT\PRINT
840 FILE #2:"PTEXT"
850 PRINT "PAEDIATRIC SUMMARY",N$,DAT$(X)
860 PRINT
870 FOR L=1 TO K-1
880 INPUT #2:T$
890 LET M=VAL(SEG$(T$,1,1))
900 LET T$=SEG$(T$,2,60)
910 IF M<>1 THEN 980
920 PRINT
930 REM ! DENOTES A NEW PARA
940 IF T$="!" THEN 970
950 REM PRINT HEADING
960 PRINT\PRINT T$\PRINT
970 LET L=L-1\GO TO 1070
980 IF A(L)<>1 THEN 1010
990 IF T$="#" THEN 1010
1000 PRINT T$
1010 FOR I=2 TO M
1020 INPUT #2:T$
1030 IF I<>A(L) THEN 1060
1040 IF T$="#" THEN 1060
1050 PRINT T$
1060 NEXT I
1070 NEXT L
1080 CLOSE #2
1090 REM STORE CODED ANSWER
1100 FILEVN #2:N$
1110 FOR L=1TO K-1\PRINT #2:A(L)\NEXT L
1120 PRINT #2:99
1130 CLOSE #2
1150 REM PRINT PERFORMANCE VARIABLES
1160 PRINT\PRINT "QUESTIONS ANSWERED 'DONT UNDERSTAND':";
1170 FOR M=1 TO J\PRINT D(M);\NEXT M
1180 PRINT\PRINT "TOTAL QUESTIONS ANSWERED:";Q
1190 END

READY
```